
Il circuito di van der Pol. 
 

Negli anni ’20 del secolo scorso, l’ingegnere elettrico olandese Balthasar van der Pol 

(1889-1959) realizzò e studiò le proprietà di un particolare circuito, usato in uno dei 

primi radioricevitori a valvole, argomento di grande attualità a quel tempo. 

Lo schema (semplificato) è ancora quello di p. 42, ma questa volta il componente 

resistivo rappresentato dal bipolo in basso è un triodo, una valvola elettronica 

costituita essenzialmente da un tubo a vuoto. Si tratta di un componente attivo (la 

circuiteria che lo pilota è omessa nello schema) in grado di erogare energia 

ogniqualvolta l’intensità della corrente che lo attraversa sia sufficientemente piccola, 

altrimenti assorbe energia; perciò la corrente tende a stabilizzarsi su un’oscillazione 

che si sostiene da sé. Più modernamente, può essere realizzato mediante dispositivi a 

semiconduttore. La sua equazione caratteristica, non lineare, ha la forma: 
 

v  =  f (i )  =  N · i
 3

  −  R · i 
 

che nel piano (i, v) è una curva come in figura: 
 

 
 

dove con  I  è indicata la radice quadrata del rapporto R / N. Quando | i | < I  (e  i  non 

è nulla) il grafico cade nel secondo e quarto quadrante, dove v·i < 0. In certi istanti, 

dunque, l’energia assorbita complessivamente dal componente potrà essere negativa, 

ossia quella erogata potrà essere positiva: ciò che caratterizza un componente attivo. 

La prima equazione di stato diventa: 
 

    (7) 
 

e quindi il sistema non è più lineare. 

Per avere anche qui un solo parametro − oltre ad assumere induttanza e capacità 

unitarie − per semplicità, e poiché questo è soltanto un esempio, possiamo pensare 

di fissare  N  =  1  / A
2
  =  1 V / A

3
,  e di studiare il comportamento del sistema 

 

    (8) 

al variare di R ( > 0). 

La prima isoclina nulla (di equazione d x/dt = 0) è la funzione y = (R − x 

2
) x = f (−x), 

che si ottiene ribaltando la caratteristica del triodo rispetto all’asse verticale; come 

varia il suo grafico al variare di R ? La seconda isoclina nulla (di equazione d y/dt = 0 

e dunque x = 0) è ancora l’asse y. La loro unica intersezione è l’origine, che rimane 

quindi il solo punto di equilibrio (però, come vedremo, instabile: è un repulsore); in 

figura sono tracciate le isocline ed è delineato il campo vettoriale nelle quattro zone 

in cui da esse è suddiviso il piano delle fasi. 
 

 
 

Tutte le soluzioni (non di equilibrio) corrispondono a traiettorie che ruotano attorno 

all’origine in senso antiorario: per saperne di più, ricorriamo come sempre a Maple. 

Fissiamo dapprima R = 1.5 . 
 

restart: 

with(DEtools): 

with(plots): 

R := 1.5: 

vdP := diff(x(t), t) = x(t)*(R-x(t)*x(t))-y(t), diff(y(t), t) = x(t): 

vars := x(t), y(t): 

tmax := 15: 

rangex := -3..3: 

rangey := -3..3: 

DEplot({vdP}, {vars}, t = 0..tmax, x = rangex, y = rangey, 

  title = `Oscillatore di van der Pol`, colour = magnitude); 

ics := [x(0)=-3, y(0)=-1], [x(0)=-3, y(0)=0], [x(0)=-3, y(0)=1], [x(0)=-3, y(0)=2], 

  [x(0)=3, y(0)=-2], [x(0)=3, y(0)=-1], [x(0)=3, y(0)=0], [x(0)=3, y(0)=1], 

  [x(0)=0.02, y(0)=0.02], [x(0)=0.02, y(0)=-0.02], [x(0)=-0.02, y(0)=0.02], 

  [x(0)=-0.02, y(0)=-0.02]: 

DEplot({vdP}, {vars}, t = 0..tmax, x = rangex, y = rangey, [ics], animate = true, 

  numframes = 300, arrows = none, thickness = 1, 

  linecolour = [blue, blue, blue, blue, blue, blue, blue, blue, red, red, red, red]); 

 

 
 

Vi è un ciclo limite attrattivo, corrispondente a un’orbita isolata (che circonda il 

punto di equilibrio instabile) verso la quale convergono tutte le traiettorie che 

partono da punti ad essa interni (eccetto il punto di equilibrio, che qui è l’origine), 

come quelle in rosso, o che partono da punti ad essa esterni, come quelle in blu. Una 

traiettoria che parta da un punto di tale orbita isolata evolverà sull’orbita stessa. 

Possiamo affermare che, mentre un punto di equilibrio stabile in un sistema 

corrisponde a un suo comportamento statico, un ciclo limite attrattivo rappresenta 

una situazione di equilibrio dinamico (asintoticamente stabile). 

Vediamo come si presenta infatti una soluzione nel tempo. 
 

corrente := DEplot({vdP}, {vars}, t = 0..1.5*tmax, x = rangex, y = rangey, 

  [[x(0)=0, y(0)=3]], stepsize = 0.01, scene = [t, x], linecolour = cyan): 

tensione := DEplot({vdP}, {vars}, t = 0..1.5*tmax, x = rangex, y = rangey, 

  [[x(0)=0, y(0)=3]], stepsize = 0.01, scene = [t, y], linecolour = magenta): 

display([corrente, tensione], 

  title = `cyan = corrente, x(0) = 0; magenta = tensione, y(0) = 3`); 

 

 



 
49 

Questo sistema mostra dunque un comportamento periodico, così come l’oscillatore 

armonico ideale, anche “in autonomia”, cioè in assenza di forze periodiche applicate 

dall’esterno. Storicamente, il termine “autonomo” era riferito proprio a quei sistemi 

di equazioni differenziali che descrivono oscillatori a valvole (tubi a vuoto) in grado 

di generare una forma d’onda periodica, senza bisogno di applicare in ingresso al 

circuito alcuna sorgente di energia a sua volta periodica (bensì costante): quello di 

van der Pol ne è un esempio. 

Ma la nostra ricerca non finisce qui: anzi, ora giungeremo a scoprire gli aspetti più 

interessanti di questo sistema. Come ormai d’abitudine, chiediamoci che cosa accade 

al variare del parametro. Lo studente provi da solo ad eseguire nuovamente la 

procedura, modificando un poco il solo valore di R (ad esempio, ponendo R:=0.7:, 

R:=1.1:, R:=1.9:, R:=2.3:) e osservando i cambiamenti (un consiglio: ingrandire 

le figure con le traiettorie ottenute con Maple, per meglio apprezzarne i dettagli). 

Si potrà notare, in prossimità dell’origine, l’allontanarsi delle traiettorie (a spirale, se 

il valore del parametro è inferiore a 2), così come la progressiva deformazione 

(allungamento e inclinazione) dell’orbita limite all’aumentare di tale valore. 

Indaghiamo dapprima su ciò che succede man mano che R aumenta. Se eseguiamo 

ancora la procedura di cui sopra, ad esempio con R:=3:, R:=4:, R:=5: (avendo 

l’accortezza di ampliare preventivamente il range di y, ponendo rangey := -6..6:), 

possiamo notare che l’orbita limite tende ad allungarsi in verticale e ad assomigliare 

a un parallelogramma dagli angoli smussati (in particolare, quelli in basso a sinistra 

e in alto a destra). Inoltre, i due “lati minori” di questo parallelogramma sono 

percorsi più velocemente degli altri due: l’andamento nel tempo della corrente 

presenta infatti salti piuttosto bruschi da valori negativi a valori positivi e viceversa. 

Proviamo allora con un incremento più significativo del parametro: 
 
restart: with(DEtools): with(plots): 

R := 20: 

vdP := diff(x(t), t) = x(t)*(R-x(t)*x(t))-y(t), diff(y(t), t) = x(t): 

vars := x(t), y(t): tmax := 20: rangex := -7..7: rangey := -40..40: 

ics := [x(0)=-7, y(0)=-20], [x(0)=7, y(0)=30], [x(0)=0, y(0)=10], [x(0)=0, y(0)=-30]: 

DEplot({vdP}, {vars}, t = 0..tmax, x = rangex, y = rangey, [ics], animate = true, 

 numframes = 200, arrows = none, thickness = 1, linecolour = [blue, blue, red, red]); 

corrente := DEplot({vdP}, {vars}, t = 0..5*tmax, x = rangex, y = rangey, 

  [[x(0)=0, y(0)=3]], stepsize = 0.01, scene = [t, x], linecolour = cyan): 

tensione := DEplot({vdP}, {vars}, t = 0..5*tmax, x = rangex, y = rangey, 

  [[x(0)=0, y(0)=3]], stepsize = 0.01, scene = [t, y], linecolour = magenta): 

display([corrente, tensione], 

  title = `cyan = corrente, x(0) = 0; magenta = tensione, y(0) = 3`); 

display(corrente); 

 

 
 

La forma dell’orbita limite si spiega facilmente sovrapponendo al quadro di stato il 

grafico della prima isoclina nulla; con Maple si può procedere nel seguente modo: 
 
p1 := phaseportrait({vdP}, {vars}, t = 0 .. tmax, [ics], x = rangex, y = rangey, 

 stepsize = 0.02, arrows = none, thickness = 1, linecolour = [blue, blue, red, red]): 

p2 := plot((R-x*x)*x, x = rangex): 

display(p1, p2); 

 

 
 

È doveroso precisare che quest’analisi vuol essere soltanto teorica; anzitutto si noti, 

dal grafico in alto a sinistra, che la tensione sul condensatore − pur partendo da un 

valore modesto e in assenza di corrente − è rapidamente soggetta a oscillazioni assai 

ampie: l’ingente quantità di energia necessaria a sostenerle dovrebbe quindi essere 

fornita dall’apparato che pilota la valvola… 

Un approccio più realistico è presentato nell’esperimento che segue. 

Riprendiamo l’equazione di stato (7), immaginiamo di fissare 
 

N  =  1  / A
2
  =  1 V / A

3
 ,    R  =  1  

 

e di variare il valore dell’induttanza L, che ora è il nostro parametro; otteniamo così 

il sistema 

 
 

con parametro L ( > 0). 

Questo sistema ha un’unica soluzione periodica (non nulla), la quale, al diminuire di 

L, tende alla curva chiusa che consiste di due segmenti orizzontali e due archi sulla 

curva y = x − x 

3
 (che è l’isoclina nulla di equazione d x/dt = 0), come in figura: 

 

 
 

Quando L è piccola (molto minore dell’unità) e quindi il suo reciproco 1/ L è grande, 

si alternano due intervalli di tempo piuttosto differenti lungo la soluzione periodica. 

Infatti, durante il movimento orizzontale, d x/dt è grande; ma d x/dt è la velocità con 

cui varia x, e quindi il percorso del ramo di traiettoria avviene rapidamente. Invece, 

in prossimità dell’isoclina nulla y = x − x 

3
, si ha d x/dt prossima a zero e d y/dt 

limitata, e quindi tale transizione è, a raffronto, assai più lenta. 

Una soluzione di questo genere fu detta da van der Pol oscillazione di rilassamento, 

e servì a dar l’avvio a una lunga serie di interessanti modelli matematici di alcuni 

fenomeni fisici, in primo luogo del battito cardiaco. 

Lasciamo allo studente l’incarico di fare una ricerca personale a tale proposito e di 

completare l’esperimento con l’ausilio di Maple, sulla falsariga di quanto abbiamo 

fatto nella presente sezione. Egli potrà convincersi, tra l’altro, del fatto che l’origine 

(comunque instabile) è un nodo quando il valore del parametro non supera 1/4. 

Se invece l’induttanza si riducesse a zero (ovvero, l’induttore fosse sostituito da un 

cortocircuito), allora l’equazione di stato − come scritta nella forma (7) − perderebbe 

validità: il sistema si ridurrebbe a una dimensione, con una sola variabile di stato (la 

tensione ai capi del condensatore, unico componente reattivo rimasto). 


